
Come Mostrare i Valori di una Misura per X Mesi 
Passati a Partire da una Varietà di Selezioni di 

Date Insieme con una Regressione 
PUBBLICATO LUGLIO 25, 2023 DI FRANCESCO BERGAMASCHI E DAVID BIANCONI 

Questo articolo è una variazione dell’ottimo articolo di SQLBI in cui hanno mostrato 
come mostrare i dati dei sei mesi precedenti rispetto alla selezione di uno slicer. Le 
differenze con questo articolo sono le seguenti: 

1. qui non useremo un calculation group (se siete interessati fatecelo 
sapere nei commenti e implementeremo questa versione); 

2. qui non duplicheremo la tabella calendario (Calendar). 

Quello che vogliamo mostrare in questo articolo è un insieme di tecniche che 
permettono all’utente di: 

• calcolare facilmente diverse date odierne (di riferimento) e sceglierne 
una (Calculated Today); 

• decidere quanti mesi indietro andare a partire dall’ultimo mese 
completo precedente quella data; 

• mostrare i valori di una misura dall’ultimo mese completo 
precedente Calculated Today fino al numero di mesi scelto 
(considerando solo i mesi interi ma questo potrebbe essere cambiato), 
insieme a una regressione lineare che mostra la tendenza. 

Non ci occuperemo della significatività statistica della regressione, per evitare che 
l’articolo diventi troppo esteso. Sul complesso argomento della significatività 
statistica di una regressione lineare potete fare riferimento a questo articolo. 

Dunque, lo scopo di questo articolo è avere una raccolta di tecniche per identificare 
un primo e un ultimo mese da mostrare in un report, in base alle selezioni dell’utente. 

Ecco il report che si vuole ottenere: 

Figura 1 

Il report è suddiviso in due sezioni, evidenziate in figura 1 da due riquadri rossi. In 
ciascuno, i passaggi numerati in figura mostrano come affrontarlo. 

Descriviamo i passaggi numerati nella figura 1. 

1. Nella sezione di destra, il rapporto è impostato per considerare oggi 
come il 4 giugno 2023 (Ref Today). Questa misura mostrerebbe, in un 
caso reale, la data reale. Il motivo per imporre una data ex ante è dovuto 
a maggiore praticità nella preparazione dell’articolo (non si sa mai 
quanto tempo ci vuole!) in modo da non dovere modificare i dati per 
mostrare i risultati nel report; 

2. uno slicer consente all’utente di scegliere tra diverse opzioni per 
impostare un’altra data (Calculated Today), che sarà il riferimento per i 
mesi passati: EOM Selected Month, EOY Selected Year, Last Fact 
Date, Today, X months Ago From Today (alcuni di questi saranno spiegati 
nelle voci seguenti in questo elenco, altri alla fine di questo elenco); 

3. il risultante Calculated Today è mostrato in una scheda; 
4. un calcolo riporta, sempre in base alla selezione al punto 2 e sempre su 

scheda, l’ultimo mese completo precedente quella data; 

5. nel caso in cui la scelta al punto 2 fosse per EOM Selected Month, 
uno slicer permetterebbe all’utente di selezionare il mese di cui vuole 
considerare l’ultima data, che diventerebbe Calculated Today; 

6. nel caso in cui la scelta al punto 2 fosse per EOY Selected Year, 
uno slicer permetterebbe all’utente di selezionare l’anno di cui vuole 
considerare l’ultima data, che diventerebbe Calculated Today (questa è 
la scelta mostrata in figura 1); 

7. nel caso in cui la scelta al punto 2 fosse per X months Ago From Today, 
uno slicer consentirebbe all’utente di selezionare il numero di mesi in 
cui desidera tornare indietro da Ref Today per trovare Calculated Today; 

8. uno slicer (Months Back) permette all’utente di scegliere il numero di 
mesi indietro partendo da Calculated Today per ottenere il primo mese 
da considerare (First Month to Consider); 

9. infine, uno slicer (Product) permette all’utente di selezionare i prodotti di 
interesse; 

10. nella sezione di sinistra della figura 1, uno slicer (Matrix Year) consente 
all’utente di selezionare l’anno o gli anni di interesse per la matrice 
sottostante. La selezione di quest’anno è indipendente da quella 
effettuata sul lato destro della figura 1 e non la filtra (ulteriori dettagli più 
avanti nell’articolo). 

Spiegazione di ulteriori scelte possibili al punto 2: nel caso in cui la scelta al punto 2 
fosse per Last Fact Date, la data massima dei fatti verrebbe calcolata 
automaticamente e diventerebbe Calculated Today; nel caso in cui la scelta al punto 
2 fosse per Today, quella data verrebbe automaticamente impostata al 4 giugno 
2023 (Ref Today) che diventerebbe (anche) Calculated Today (di nuovo, questa data 
sarebbe, nel mondo reale, la data del giorno reale). 

Il grafico a linee, basato sulle scelte dell’utente, mostra la misura Total Amount for X 
Past Months (totally dynamic) e una regressione dei valori di quella misura nei mesi 
coerenti con le scelte dell’utente (dal primo mese da considerare fino all’ultimo 
mese completo, in altre parole sempre su mesi interi, ma questo potrebbe essere 
facilmente modificato per includere periodi all’interno di un mese). 

L’uso classico di questo report è il seguente: supponiamo che l’interesse sia 
osservare la misura Total Amount per il prodotto Product2 nell’anno 2023 sul lato 
sinistro, potendo osservarla anche negli ultimi 9 mesi a partire dalla fine dell’anno 
precedente, 2022. Questo è esattamente ciò che mostra la figura 1. E le modifiche 
possono essere facilmente apportate tramite gli slicer. 

Il modello utilizzato è un semplice schema a stella, visibile di seguito. 

Figura 2 

 
Sviluppo 

Di seguito le misure e/o le tabelle realizzate. Andremo nello stesso ordine 
dell’elenco nella sezione introduttiva, così sarà più facile capire come funziona il 
report. 

Ref Today = 
 
DATE ( 2023, 6, 4 ) 
— il codice, in un reale applicazione, sarebbe TODAY ( ) 

Le diverse opzioni per impostare un’altra data vengono mostrate in uno slicer tramite 
una tabella importata creata in Power Query (Calculated Today Selection). 

Figura 3 

 
 
Calculated Today = 
VAR MonthsAgoSelection = [Months Ago Value] 
VAR EOMSelection = [End Of Month] 
VAR SelectedYear = 
    SELECTEDVALUE ( ‘Year'[Year Selection] ) 
RETURN 
    SWITCH ( 
        SELECTEDVALUE ( ‘Calculated Today'[Calculated Today Selection] ), 
        “Today”, DATE ( 2023, 6, 4 ), 
        — Should have been TODAY() 
        “Last Fact Date”, 
            CALCULATE ( 
                MAX ( Facts[Date] ), 
                REMOVEFILTERS ( ‘Calendar’ ) 
            ), 
        “X Months Ago From Today”, 
            CALCULATE ( 
                MINX ( 
                    DATESINPERIOD ( 
                        ‘Calendar'[Date], 
                        DATE ( 2023, 6, 4 ), 
                        – MonthsAgoSelection, 
                        MONTH 
                    ), 
                    [Date] 
                ), 
                REMOVEFILTERS ( ‘Calendar’ ) 
            ), 

        “EOM Selected Month”, EOMSelection, 
        “EOY Selected Year”, DATE ( SelectedYear, 12, 31 ) 
    ) 
 
 
Calculated Last Complete Month = 
VAR TodaySelection = [Calculated Today] 
RETURN 
    CALCULATE ( 
        MAXX ( 
            FILTER ( 
                ADDCOLUMNS ( 
                    ALL ( ‘Calendar'[Year-Month] ), 
                    “@EOM”, 
                        CALCULATE ( MAX ( ‘Calendar'[Date] ) ) 
                ), 
                [@EOM] <= TodaySelection 
            ), 
            ‘Calendar'[Year-Month] 
        ), 
        REMOVEFILTERS ( ‘Calendar’ ) 
    ) 

I valori visualizzati nello slicer EOM Selected Month provengono da una tabella 
calcolata (End Of Month). 

End Of Month = 
ALLNOBLANKROW ( ‘Calendar'[End of Month] ) 

Una misura (End Of Month) acquisisce la selezione dell’utente. 

End Of Month = 
SELECTEDVALUE ( ‘End Of Month'[End of Month Selection] ) 

Nel suddetto filtro dei dati viene utilizzata una misura come filtro (Prior to Last 
Complete Month), con il valore 1. 

Prior to Last Complete Month = 
INT ( 
    FORMAT ( 

        MAX ( ‘End Of Month'[End of Month Selection] ), 
        “YYYYmm” 
    ) <= [Calculated Last Complete Month To Filter] 
) 
 
 
Calculated Last Complete Month To Filter = 
CALCULATE ( 
    MAXX ( 
        FILTER ( 
            ADDCOLUMNS ( 
                ALL ( ‘Calendar'[Year-Month] ), 
                “@EOM”, 
                    CALCULATE ( MAX ( ‘Calendar'[Date] ) ) 
            ), 
            [Total Amount] > 0 
        ), 
        ‘Calendar'[Year-Month] 
    ), 
    REMOVEFILTERS ( ‘Calendar’ ) 
) 

Il filtro sopra riportato evita di mostrare i mesi in cui non ci sono fatti. 

I valori mostrati nello slicer EOY Selected Year provengono da una tabella calcolata 
(Year). 

Year = 
ALLNOBLANKROW ( ‘Calendar'[Year] ) 

I valori mostrati nello slicer X months Ago From Today provengono da una tabella 
calcolata collegata a un parametro campo (Months Ago). 

Months ago = 
GENERATESERIES ( 1, 12, 1 ) 

Una misura (Month Ago Value) acquisisce la selezione dell’utente. 

Months Ago Value = 
SELECTEDVALUE ( ‘Months ago'[Months Ago Selection] ) 

I valori visualizzati nello slicer Months Back provengono da una tabella calcolata 
collegata a un parametro campo (Months Back). 

Months Back = 
GENERATESERIES ( 3, 12, 3 ) 

Una misura (Months Back) acquisisce la selezione dell’utente. 

Months Back = 
SELECTEDVALUE ( ‘Months Back'[Months Back Selection], 12 ) 

I valori mostrati nello slicer Matrix Year provengono dalla colonna Calendar[Year]. Il 
calendario è una tabella calcolata. 

Calendar = 
VAR MinDate = 
    DATE ( YEAR ( MIN ( Facts[Date] ) ), 1, 1 ) 
VAR MaxDate = 
    DATE ( YEAR ( MAX ( Facts[Date] ) ), 12, 31 ) 
RETURN 
    ADDCOLUMNS ( 
        CALENDAR ( MinDate, MaxDate ), 
        “Year-Month”, 
            YEAR ( [Date] ) & FORMAT ( MONTH ( [Date] ), “00” ), 
        “Year”, YEAR ( [Date] ), 
        “End of Month”, 
            EOMONTH ( [Date], 0 ) 
    ) 
 
 
Total Amount = 
SUM ( Facts[Amount] ) 
 
 
Total Amount for X Past Months (totally dynamic) = 
VAR TodaySelection = [Calculated Today] 
VAR MonthsBackSelection = [Months Back] 

VAR EOMSelection = [End Of Month] 
VAR LastCompleteMonth = 
    CALCULATE ( 
        MAXX ( 
            FILTER ( 
                ADDCOLUMNS ( 
                    ALL ( ‘Calendar'[Year-Month] ), 
                    “@EOM”, 
                        CALCULATE ( 
                            MAX ( ‘Calendar'[Date] ) 
                        ) 
                ), 
                [@EOM] <= TodaySelection 
            ), 
            ‘Calendar'[Year-Month] 
        ), 
        REMOVEFILTERS ( ‘Calendar’ ) 
    ) 
VAR FirstMonthToConsider = 
    MAXX ( 
        CALCULATETABLE ( 
            OFFSET ( 
                – MonthsBackSelection + 1, 
                FILTER ( 
                    ALL ( ‘Calendar'[Year-Month] ), 
                    ‘Calendar'[Year-Month] <= LastCompleteMonth 
                ), 
                ORDERBY ( ‘Calendar'[Year-Month] ) 
            ), 
            REMOVEFILTERS ( ‘Calendar’ ) 
        ), 
        ‘Calendar'[Year-Month] 
    ) 
RETURN 
    CALCULATE ( 
        [Total Amount], 
        KEEPFILTERS ( ‘Calendar'[Year-Month] >= FirstMonthToConsider 
            && ‘Calendar'[Year-Month] <= LastCompleteMonth ) 
    ) 

Linear Regression Total Amount = 
VAR TodaySelection = [Calculated Today] 
VAR MonthsBackSelection = [Months Back] 
VAR EOMSelection = [End Of Month] 
VAR LastCompleteMonth = 
    CALCULATE ( 
        MAXX ( 
            FILTER ( 
                ADDCOLUMNS ( 
                    ALL ( ‘Calendar'[Year-Month] ), 
                    “@EOM”, 
                        CALCULATE ( 
                            MAX ( ‘Calendar'[Date] ) 
                        ) 
                ), 
                [@EOM] <= TodaySelection 
            ), 
            ‘Calendar'[Year-Month] 
        ), 
        REMOVEFILTERS ( ‘Calendar’ ) 
    ) 
VAR FirstMonthToConsider = 
    MAXX ( 
        CALCULATETABLE ( 
            OFFSET ( 
                – MonthsBackSelection + 1, 
                FILTER ( 
                    ALL ( ‘Calendar'[Year-Month] ), 
                    ‘Calendar'[Year-Month] <= LastCompleteMonth 
                ), 
                ORDERBY ( ‘Calendar'[Year-Month] ) 
            ), 
            REMOVEFILTERS ( ‘Calendar’ ) 
        ), 
        ‘Calendar'[Year-Month] 
    ) 
RETURN 
    IF ( 
        [Total Amount for X Past Months (totally dynamic)] <> 0, 
        IF ( 
            ISINSCOPE ( ‘Calendar'[End of Month] ), 
            VAR line = 

                LINESTX ( 
                    FILTER ( 
                        ALLSELECTED ( 
                            ‘Calendar'[End of Month], 
                            ‘Calendar'[Year-Month] 
                        ), 
                        ‘Calendar'[Year-Month] >= FirstMonthToConsider 
                            && ‘Calendar'[Year-Month] < LastCompleteMonth 
                    ), 
                    [Total Amount for X Past Months (totally dinamic)], 
                    ‘Calendar'[End of Month] 
                ) 
            VAR slope = 
                SELECTCOLUMNS ( 
                    line, 
                    [Slope1] 
                ) 
            VAR intercept = 
                SELECTCOLUMNS ( 
                    line, 
                    [Intercept] 
                ) 
            VAR x = 
                SELECTEDVALUE ( ‘Calendar'[End of Month] ) 
            VAR y = x * slope + intercept 
            RETURN 
                y, 
            SUMX ( 
                ALLSELECTED ( 
                    ‘Calendar'[End of Month], 
                    ‘Calendar'[Year-Month] 
                ), 
                [Total Amount for X Past Months (totally dinamic)] 
            ) 
        ) 
    ) 

Un punto importante è che il filtro applicato dallo slicer Matrix Year non deve 
influenzare il grafico a linee. Ciò si ottiene tramite la funzionalità Modifica interazioni 
di Power BI Desktop, come mostrato nella figura 4. 

Figura 4 

Grazie all’interazione modificata sopra, non è necessario duplicare la 
tabella Calendar. 

Conclusioni 

Come variazione del grande articolo di SQLBI, in questo articolo ci sono diverse 
possibili configurazioni che l’utente può creare per scegliere facilmente un intervallo 
di mesi in cui mostrare una misura. Riteniamo che sarebbe un buon esercizio 
replicare tutte le opzioni fornite all’utente in questo articolo, al fine di crescere sui 
calcoli DAX e sui contesti di valutazione. Un buon report è un buon report se questa 
è l’opinione degli utenti. Quindi è in base alle loro esigenze che dovremmo agire, 
rendendo la loro vita il più semplice possibile per ottenere ciò che vogliono 
osservare. 

